

A SPICE MODEL FOR TRIACs *(preliminary version; incomplete)*

A. F. Petrie, Independent Consultant, 7 W. Lillian Ave., Arlington Heights, IL 60004

Charles Hymowitz, Intusoft, 222 West 6th St. Suite 1070, San Pedro, CA 90731, (310) 833-0710, FAX (310) 833-9658, E-mail 74774.2023@compuserve.com

SPICE is the most popular program for simulating the behavior of electronic circuits. The biggest stumbling block that engineers run into is turning vendor data sheet specifications into SPICE models that emulate real devices and run without convergence problems. This is especially true for power devices, like Triacs, where the cost of testing and possibly destroying devices is prohibitive. The following paper describes the FIRST known SPICE subcircuit macro model for a Triac[1].

Introduction

Berkeley SPICE is the most popular program for simulating the behavior of electronic circuits. The biggest stumbling block most engineers run into is turning company specifications or vendor data sheets into SPICE device models that emulate real devices and run without problems. Here is an approach which works well for an TRIAC and derives directly from data book parameters.

A triac is a bi-lateral switch that can be triggered into conduction regardless of its polarity. It is modeled by using two NPN/PNP transistor pairs connected back-to-back as shown on the left of Figure 4. The base of each transistor is connected to the collector of the other. This produces positive feedback, resulting in the required switching action. At MT2, the emitter metallization overlaps the base of the NPN transistor. This forms a lateral resistor in the base (P) region and is shown as RH in the model. RH

determines the holding current of the triac. A similar resistor exists at MT1. These resistors hold the triac off unless triggered by the gate or the holding current from a previous "on" condition.

Although this configuration has the basic triggering function of the triac, it must be enhanced to emulate other important parameters such as off state leakage, breakover voltage and current, and voltage and current characteristics in all four modes of operation. Figure 4 shows the full triac subcircuit. Resistors and zener diodes are used to simulate the breakdown voltages and leakage currents. Dependent sources are used to emulate the various triggering modes and to allow a wider range of trigger data to be entered.

If adjusting the model parameters for the triac subcircuit seems a little daunting, you can use SPICEMOD. SPICEMOD is a software program that quickly converts data book parameters into SPICE model parameters. Entering only the device type and maximum voltage and current ratings will produce a realistic model as all other parameters are




Figure 1. Basic Triac structure




Figure 2. Basic Triac subcircuit




Figure 3

scaled from them. Naturally, the more data you enter, the more exact the model will be. Figure 5 shows the triac entry screen from SPICEMOD.

This model and method are used for the TRIAC in the SPICEMOD (v2.3) program. This program allows direct conversion of data book information to SPICE models for diodes, zeners, bipolar transistors, JFETs, MOSFETs, SCRs and now Triacs. Models can be generated as fast as data can be entered. This makes it easy to generate models for custom or "house" devices. Limit devices can be modeled by entering limit device data. Because these models use standard SPICE syntax, they can be used with any program that adheres to SPICE rules.

Data books may show the same values for all four gate trigger current and voltage modes. However, the values are different in all four modes. In a real triac, when the gate is driven with the opposite polarity from the MT2 terminal, the gate must supply enough current to drive the shunting resistance to twice the gate voltage, thus the trigger current cannot be the same for both polarities, regardless of what the data book limits say. G+ with MT2- always requires more gate drive than other modes. In this case, it is only necessary to enter the first mode gate voltage and current and let the program estimate the others for the most realistic model. SPICEMOD guards against entering grossly unrealistic data and will attempt to produce the closest realistic model in these situations.

The SPICE 2G compatible netlist for the triac is shown in Table 1 (next page). Along with the "Affects" column in




Figure 2. SPICE 2G.6 compatible Triac subcircuit

Figure 5 you can get an idea of the relationship between which data sheet parameters correspond to which SPICE model parameters.

Table 1. 2N5568 Triac Subcircuit

```

.SUBCKT 2N5568 1 2 3
* TERMINALS: MT2 G MT1 Mot. 400V 10A
QN1 5 4 3 NOUT ; Output Transistors
QN2 11 6 7 NOUT ; "OFF" keyword for Q devices
QP1 6 11 3 POUT
QP2 4 5 7 POUT
DF 4 5 DZ ; Forward breakdown diode
DR 6 11 DZ ; Reverse breakdown diode
RF 4 6 40MEG ; Forward leakage current (controls IDRM).
RT2 1 7 52.8M ; Controls "on" resistance
RH 7 6 75 ; Controls reverse holding current
RGP 8 3 54.5 ; Controls forward holding current and trigger
current
RG 2 8 26.4 ; with RGP controls VGT
RS 8 4 52.6 ; with RGP controls forward holding current
DN 9 2 DIN ; Diode to isolate G- triggering modes
RN 9 3 27.8 ; Controls current in G- trigger modes
GNN 6 7 9 3 38.8M ; Controls G-, MT2- trigger voltages
GNP 4 5 9 3 51.2M ; Controls G-, MT2+ trigger voltages
DP 2 10 DIP ; Diode to isolate G+, MT2- trigger mode
RP 10 3 16.2 ; Controls current in G+, MT2- trigger mode
GP 7 6 10 3 26.1M ; Controls G+, MT2- trigger voltage
.MODEL DIN D (IS=53.5F) ; Conducts in G- modes
.MODEL DIP D (IS=53.5F N=1.19) ; Higher drop diode conducts only in
G+, MT2- mode
.MODEL DZ D (IS=53.5F N=1.5 IBV=10U BV=400)
.MODEL POUT PNP (IS=53.5F BF=5 CJE=235P TF=25.5U)
.MODEL NOUT NPN (IS=53.5F BF=20 CJE=235P CJC=46.9P TF=1.7U)
.ENDS

```

Table 1, SPICE 2G Triac model generated by the SPICEMOD modeling program from manufacturer's data sheet parameters.




Figure 3, Data sheet parameters (above left) used to create the SPICE TRIAC subcircuit (Table 1). To make a new model, data sheet values are entered into the SpiceMod entry screen. As they are entered the subcircuit values are calculated. The more data that is entered, the more accurate the final model will be. The subcircuit parameters affected by each entered parameter are shown to right.

Figure 4 shows the SPICE subcircuit that I developed to solve these problems. Replacing the trigger transistors with diodes and voltage controlled current sources reduces the number of SPICE parameters needed and provides some isolation between the four modes of operation. The gate connections still emulate the feedback to the gate.

Table 1 lists the .SUBCKT for this same model. QN1, an NPN transistor, has a forward beta of 20 and is connected to the gate at its base. QP2, a PNP transistor, is connected to MT2 and QN1 in a positive feedback mode. Due to the fact that there are two stable states (on and off) for the TRIAC, it is necessary to add the “OFF” statement after the transistors in the .SUBCKT model or SPICE may never evaluate the off condition or may hang due to the

uncertainty. The OFF command causes SPICE to start the calculations assuming the off condition. If the device is really “on”, there is no stable “off” state, so the OFF command has no effect on the accuracy of the result. (Hint: any time SPICE is used to evaluate a bi-stable device, the OFF or ON command must be used to obtain the correct state.)

Triacs have two stable states. Therefore, it may be necessary to tell SPICE which state to use, especially when you want to start a simulation with the triac in the off state. To do this you can issue the “OFF” keyword on the subcircuit transistor lines.

Figure 6 shows the low current region of a Motorola 2N5568 triac with no input applied to the gate. This curve was generated by holding the current to the gate at zero and sweeping the MT2 current from -50mA to +50mA. The slope of the curve at the zero axis is determined by the IDR_M specification. The maximum voltage swings in the off state are determined by the VDR_M specification. The trigger points (+6 and -10mA) are determined by the holding current.

Simplified behavioral triac models are included with IsSPICE4. Their superior simulation speed and idealized response can make them useful for investigating triac control circuitry. However, with power devices such as the triac, you will normally need models that exhibit 2nd order effects as well. Without them you can't run realistic




Figure 6, DC analysis of the Motorola 2N5568 triac. Several FREE models are posted on the Intusoft Web Site.

simulations. Most other SPICE vendors do not offer such sophisticated power semiconductor models. Fortunately, Intusoft does.

Figure 5 is a SPICE plot of the low current region of a Motorola 2N5568 TRIAC with no input applied to the gate. This curve was generated by holding the current to the gate at zero and sweeping the MT2 current from -50 mA. to +50 mA. The slope of the curve at the zero axis is determined by the IDR_M specification. The maximum voltage swings in the off state are determined by the VDR_M specification. The trigger points (+ 6 and -10 mA. in this case) are determined by the holding current specification (controlled by resistors R_S, R_{GP} and R_H in figure 3). Table 2 shows the SPICE file used to generate this analysis. (The subcircuit is brought in with the *INCLUDE statement in IS-SPICE.)

There are four gate trigger modes that must be modeled. The first mode is similar to an SCR with MT2 positive and the gate also positive. For convenience, this is combined with the fourth mode where the gate is also positive and MT2 is negative. Figure 6 is a plot of MT2 voltage verses gate current using a 12 volt supply and a 100 ohm load resistor. Note that the device turns on at the specified 11 mA. when MT2 is positive and at 31 mA. of gate current when MT2 is negative. The first mode gate current and voltage are controlled by R_G and R_{GP} in figure 4. The fourth mode gate current and voltage are controlled by resistor R_E and the forward beta (BF) of Q_{PG}. Table 3 lists

```


TRIAC-GT.CIR - TRIAC GATE TRIGGER CURVES
.OPTIONS NOMOD NOPAGE RELTOL=.01 ITL1=500 ITL2=1500
*.DC IG 0 -.05 -.0005 VCC -12 12 24
* ^ Gate Trigger Curve
* Note: for some triac models the DC analysis will
* not converge. The Transient analysis may be substituted
.TRAN 1u 1m UIC
.PRINT DC V(1) V(2)
.PRINT TRAN V(1) V(2)
*ALIAS V(1)=MT2 VOLTAGE
*ALIAS V(2)=GATE VOLTAGE
*ALIAS IT2=MT2 CURRENT
*ALIAS IG=GATE CURRENT
*INCLUDE TRIAC45.LIB
*
* ^ Use your library path and name.
VCC 3 0 -12V ; or 12V
*
* ^Set for the proper biasing if using transient
RL 3 1 100
*
* ^Adjust for device.
IG 0 2 0 pulse 0 -.05 0 1M
X1 1 2 0 T435-800 ; 2N5568
*
* ^ Change to desired device.
.END

```

Figure 6. The switching circuit TRIAC-GT.CIR used to test the transient Triac performance.

the SPICE file used to generate this curve.

Figure 7 shows a similar curve for the Gate negative mode of operation. This curve was generated by changing IG to negative values in the .DC command in Table 3. This triggering mode makes use of diode DN, resistor RN, and transconductance generator GN in the subcircuit to turn on transistors QP1 or QP2. GN is used to adjust the gate sensitivity when MT2 is negative and BF of QP2 is used to adjust the gate sensitivity when MT2 is positive. This method allows adjustment of gate sensitivity with minimum interaction. Note that an NPN transistor could have



been used in place of DN, RN and GN in the model, but a transistor model is more complex and the control parameters more difficult to calculate.

SPICE Problems:

When running a SPICE analysis using a bi-stable device like a TRIAC, the device usually switches on or off, causing a discontinuity in the output. Since SPICE makes new calculations starting with the previous values, it doesn't like discontinuities. To help SPICE converge, the following OPTIONS statement is recommended in the .CIR file:

```
.OPTIONS RELTOL=.01 ITL1=500 ITL2=200
```

The increased iterations only occur when needed and the reduced accuracy is adequate for a switching circuit. This is far more desirable than receiving a "no convergence" message.

I made mention earlier of the unrealistic data found in manufacturer's data books. In a real Triac, when the gate is driven with the opposite polarity from the MT2 terminal, the gate must supply enough current to drive the shunting resistance to twice the gate voltage, thus the trigger current cannot be the same for both polarities, regardless of what the data book limits say. The values are just made high enough to cover all devices and polarities. For this reason, the SpiceMod program will not allow grossly unrealistic values to be entered.

References:

1. R. L. Avant, F. C. Lee, and D. Y. Chen, A Practical SCR Model for Computer Aided Analysis of AC Resonant Charging Circuits, IEEE, July, 1981.
2. Paolo Antognetti and Giuseppe Massobrio, Semiconductor Device Modeling with SPICE, McGraw-Hill, 1988.
3. "Thyristor Device Data," Motorola Inc., DL131, REV3, 1991.

SPICEMOD v2.3 models diodes, Bjts, Jfets, Mosfets, SCRs, IGBTs, Power Mos, Power BJTs, and Darlingtons BJTs and now models zeners and triacs. SPICEMOD is compatible with all SPICE simulators and is available immediately for \$300. Upgrades are \$50. Thanks to the author and developer of SPICEMOD, Slim Petrie, 7 W. Lillian Ave., Arlington Heights, IL 60004.

A. F. "Slim" Petrie is retired from Motorola Inc., where he was a Principal Staff Engineer and Dan Noble Fellow. He develops both hardware and software and holds 24 U S Patents. His "Circuit Analysis" program was sold through Apple Computer in the early 1980s. With a keen appreciation for the problems of the working engineer, he continues to develop tools to make that job easier. He can be reached at 7 W. Lillian Ave., Arlington Heights, IL 60004, where he welcomes your comments.